
	Reason for change:
	Test case fails at step 33 per prose for the following reasons:

1) The testcase fails at 34.229.Annex C.31 step 1 - Re-INVITE PDU “reason parameter” mismatch due to the impementation of R5-147099. Current prose expects UE to set reason-text value to "handover cancelled" or “failure to transition to CS domain” regardless of the UE’s release capabilty. However, this condition should only be applicable to the release >= 10 capable UEs.

2) The testcase fails at 34.229.Annex C.31 step 1 - Re-INVITE PDU “CSeq parameter” mismatch due to incorrect TTCN implementation. CSeq generation/increment is not required for the first UE initiated transaction PDU.



	
	

	Summary of change:
	1) Implement UE release capabily version dependant reason text check logics for the Re-INVITE PDU in function f_IMS_InviteRequest_MessageHeaderRX
2) Replace CSeq increment function f_IMS_PTC_ImsInfo_CseqIncr in function f_IMS_InviteRequest_MessageHeaderRX with wildcard for any value statement instead.

	Other comments:
	


Change 1
	Function name
	function f_IMS_SRVCC_ReInvite_AnnexC31_Step1

	Reason for change
	The testcase fails in 34.229.Annex C.31 step 1 - Re-INVITE PDU “reason parameter” mismatch due to the impementation of R5-147099. Current prose expects UE to set reason-text value to "handover cancelled" or “failure to transition to CS domain” regardless of the UE’s release capabilty. However, this condition should only be applicable to the release >=10 capable UEs.

Note: A draft prose CR is required and will be presented at RAN5#63 meeting

	Summary of change
	Implement UE release capabily version dependant reason text check logics for the Re-INVITE PDU in function f_IMS_InviteRequest_MessageHeaderRX


	TTCN module
	TTCN\Common\IMS\IMS_Procedures_CallControl.ttcn

	MCC160 Comment
	Accepted


Before change

	  function f_IMS_SRVCC_ReInvite_AnnexC31_Step1(boolean p_GeolocationInfoAvailable := false)

    runs on IMS_PTC return IMS_DATA_REQ

  { /* @sic R5s140020, R5-140799: new function sic@ */

    /* @sic R5-140799 sic@ */

    var template (omit) float v_WaitDuration := omit;   // no wait duration

    var template (omit) integer v_UE_Release := f_IMS_PTC_GetUE_Release();

    var template (present) SDP_Message v_SDP_MessageRX := f_IMS_BuildSDP_AnnexC31_Step1();

    var template (present) MessageBody v_MessageBody := f_SIP_MessageBodySDP_RX(v_SDP_MessageRX, p_GeolocationInfoAvailable);

    var template (present) MessageHeader v_MessageHeader := cr_MessageHeader_Dummy;

    var template charstring v_ReasonText := ("failure to transition to CS domain", "handover cancelled");

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var SDP_Message v_SDP_Message;
    /*

    if (not (ispresent(v_UE_Release) and (valueof(v_UE_Release) >= 11))) {

       v_ReasonText := omit;

    }
    */

    v_MessageHeader.reason := cr_Reason(487, v_ReasonText);     // @sic R5-140799 sic@

    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_Common(A_2_1_A5, v_MessageHeader, -, v_MessageBody, -, v_WaitDuration, -, p_GeolocationInfoAvailable));

    v_SDP_Message := f_IMS_MessageBody_CheckGeolocationAndGetSDP(v_IMS_DATA_REQ.Request.Invite.messageBody, p_GeolocationInfoAvailable);

    // check at least one connections field sent

    f_IMS_SDP_CheckConnections(v_SDP_Message);

    return v_IMS_DATA_REQ;

  }


After change

	  function f_IMS_SRVCC_ReInvite_AnnexC31_Step1(boolean p_GeolocationInfoAvailable := false)

    runs on IMS_PTC return IMS_DATA_REQ

  { /* @sic R5s140020, R5-140799: new function sic@ */

    /* @sic R5-140799 sic@ */

    var template (omit) float v_WaitDuration := omit;   // no wait duration

    var template (omit) integer v_UE_Release := f_IMS_PTC_GetUE_Release();

    var template (present) SDP_Message v_SDP_MessageRX := f_IMS_BuildSDP_AnnexC31_Step1();

    var template (present) MessageBody v_MessageBody := f_SIP_MessageBodySDP_RX(v_SDP_MessageRX, p_GeolocationInfoAvailable);

    var template (present) MessageHeader v_MessageHeader := cr_MessageHeader_Dummy;

    var template charstring v_ReasonText := ("failure to transition to CS domain", "handover cancelled");

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var SDP_Message v_SDP_Message;

    // Anritsu_13436 R5-147099 update skip the v_ReasonText check for UE release >=10 

    if (not (ispresent(v_UE_Release) and (valueof(v_UE_Release) >= 10))) {

       v_ReasonText := omit;

    }    
    v_MessageHeader.reason := cr_Reason(487, v_ReasonText);     // @sic R5-140799 sic@

    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_Common(A_2_1_A5, v_MessageHeader, -, v_MessageBody, -, v_WaitDuration, -, p_GeolocationInfoAvailable));

    v_SDP_Message := f_IMS_MessageBody_CheckGeolocationAndGetSDP(v_IMS_DATA_REQ.Request.Invite.messageBody, p_GeolocationInfoAvailable);

    // check at least one connections field sent

    f_IMS_SDP_CheckConnections(v_SDP_Message);

    return v_IMS_DATA_REQ;

  }
	


Change 2
	Function name
	f_IMS_InviteRequest_MessageHeaderRX

	Reason for change
	The testcase fails at 34.229.Annex C.31 step 1 - Re-INVITE PDU “CSeq parameter” mismatch due to incorrect TTCN implementation. CSeq generation/increment is not required for the first UE initiated transaction PDU.



	Summary of change
	Replace CSeq increment function f_IMS_PTC_ImsInfo_CseqIncr in function f_IMS_InviteRequest_MessageHeaderRX with wildcard for any value statement instead.

	TTCN module
	TTCN\Common\IMS\IMS_Procedures_CallControl.ttcn

	MCC160 Comment
	Accepted in principle:
TTCN can know about whether there has been a remote CSeq already or whether any value needs to be allowed (see MCC160 Implementation below).

NOTE: the proposed implementation would miss test requirements e.g. in case of sub-sequent re-INVITE


Before change

	  function f_IMS_InviteRequest_MessageHeaderRX(template (present) MessageHeader p_MessageHeader := cr_MessageHeader_Dummy,

                                               template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := false,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  { /* p_A3:          MTSI
    :

    :
    var OptionTag_List v_OptionTag_List := tsc_OptionTagList_100rel & p_AdditionalOptionTagsForSupported;

    if (p_Context == A_2_1_A6) {  // INVITE for creating an emergency session in case of no registration

      // Via, Route, Contact: ports to be used

      v_Port_us := ?;

      v_Port_ps := tsc_IMS_PortNumber_5060;

      // From: Display

      v_Displayname := cr_QuotedStringOrToken("Anonymous");     /* @sic R5s140005, R5-140901 sic@ */

    }

    else {

      // Via, Route, Contact: ports to be used

      if (not v_IsGiba or (p_Context == A_2_1_A7)) {

        v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

        v_Port_us := v_Protected.Port_us;

        v_Port_ps := v_Protected.Port_ps;

      }

    }

    if (p_Context == A_2_1_A5)  {                  /* re-invite @sic R5s140017 MCC160 implementation sic@ */

      v_RouteBodyList := fl_RouteSet_RouteBodyList_RX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

      v_From := f_IMS_RequestInDialog_FromHeaderRX();

      v_To := f_IMS_RequestInDialog_ToHeaderRX();

      v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogRemote);
      v_SessionId := f_IMS_PTC_ImsInfo_DialogGetSessionId();

    }


After change

	  function f_IMS_InviteRequest_MessageHeaderRX(template (present) MessageHeader p_MessageHeader := cr_MessageHeader_Dummy,

                                               template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := false,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  { /* p_A3:          MTSI
    :

    :
    var OptionTag_List v_OptionTag_List := tsc_OptionTagList_100rel & p_AdditionalOptionTagsForSupported;

    if (p_Context == A_2_1_A6) {  // INVITE for creating an emergency session in case of no registration

      // Via, Route, Contact: ports to be used

      v_Port_us := ?;

      v_Port_ps := tsc_IMS_PortNumber_5060;

      // From: Display

      v_Displayname := cr_QuotedStringOrToken("Anonymous");     /* @sic R5s140005, R5-140901 sic@ */

    }

    else {

      // Via, Route, Contact: ports to be used

      if (not v_IsGiba or (p_Context == A_2_1_A7)) {

        v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

        v_Port_us := v_Protected.Port_us;

        v_Port_ps := v_Protected.Port_ps;

      }

    }

    if (p_Context == A_2_1_A5)  {                  /* re-invite @sic R5s140017 MCC160 implementation sic@ */

      v_RouteBodyList := fl_RouteSet_RouteBodyList_RX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

      v_From := f_IMS_RequestInDialog_FromHeaderRX();

      v_To := f_IMS_RequestInDialog_ToHeaderRX();

      // Anritsu_13436 - UE initiated transaction
      // v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogRemote);

      v_CSeqValue := ?;
      v_SessionId := f_IMS_PTC_ImsInfo_DialogGetSessionId();

    }
	


MCC160 Implementation
	  function f_IMS_InviteRequest_MessageHeaderRX(template (present) MessageHeader p_MessageHeader := cr_MessageHeader_Dummy,

                                               template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := false,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  {
...

    if (p_Context == A_2_1_A5)  {                  /* re-invite @sic R5s140017 MCC160 implementation sic@ */

      v_RouteBodyList := fl_RouteSet_RouteBodyList_RX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

      v_From := f_IMS_RequestInDialog_FromHeaderRX();

      v_To := f_IMS_RequestInDialog_ToHeaderRX();

      if (isvalue(f_IMS_PTC_ImsInfo_CseqGet(dialogRemote))) {  /* @sic R5s140235 sic@ */

        v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogRemote);

      }
      v_SessionId := f_IMS_PTC_ImsInfo_DialogGetSessionId();

    }
...

  }


	  function f_IMS_PTC_ImsInfo_CseqGet(IMS_PTC_CseqType_Type p_CseqType) runs on IMS_PTC return template (omit) integer

  { /* @sic R5s140235: return value changed to "template (omit)" sic@ */

    var IMS_PTC_DialogInfo_Type v_Dialog := fl_IMS_PTC_ImsInfo_GetCurrentDialog();

    var template (omit) integer v_CseqValue;

    select (p_CseqType) {

      case (register)     { v_CseqValue := vc_IMS_Global.ImsInfo.Registration.Cseq; }

      case (dialogRemote) { v_CseqValue := v_Dialog.RemoteCseq; }

      case (dialogLocal)  { v_CseqValue := v_Dialog.LocalCseq; }

      case else { FatalError (__FILE__, __LINE__, ""); }

    }

    return v_CseqValue;

  }


	  function f_IMS_PTC_ImsInfo_CseqGetValue(IMS_PTC_CseqType_Type p_CseqType) runs on IMS_PTC return integer

  { /* @sic R5s140235: new function sic@ */

    var template (omit) integer v_Cseq := f_IMS_PTC_ImsInfo_CseqGet(p_CseqType);

    if (not isvalue(v_Cseq)) {

      FatalError(__FILE__, __LINE__, "no valid Cseq yet");

    }

    return valueof(v_Cseq);

  }


	  function f_IMS_PTC_ImsInfo_CseqIncr(IMS_PTC_CseqType_Type p_CseqType) runs on IMS_PTC return integer

  { /* @sic R5s140235: Cseq may be omit sic@ */

    var integer v_CseqValue := f_IMS_PTC_ImsInfo_CseqGetValue(p_CseqType) + 1;

    f_IMS_PTC_ImsInfo_CseqSet(p_CseqType, v_CseqValue);

    return v_CseqValue;

  }


	  type record IMS_PTC_DialogInfo_Type {                 /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT)

                                                           @sic R5s140092 change 1.5: RemoteURI, RemoteCseq, LocalURI, LocalCseq stored in this structure sic@ */

    boolean             MobileOriginated  optional,     /* needed e.g. to get the right route-set for re-INVITE */

    charstring          RemoteTag   optional,           /* acc. to RFC 3261 when creating a dialog in first request there is a From tag but no To tag;

                                                           the To tag is chosen by the other send and sent in a response; in all further messages within the dialogthe same From and To tags shall be used

                                                           (=> especially in request messages (e.g. BYE) it may be necessary to use this stored value whereas in responses the values can be used as sent in the request);

                                                           NOTE: in general outside a dialog (e.g. registration) requests get a new From tag and the corresponding reponses contain a new To tag */

    charstring          RemoteURI   optional,           /* URI of the remote side in a dialog created by INVITE */

    integer             RemoteCseq  optional,           /* @sic R5s140235: changed to optional sic@ */

    charstring          LocalTag    optional,

    charstring          LocalURI    optional,           /* URI of the UE in a dialog created by INVITE */

    integer             LocalCseq,

    charstring          CallId      optional,

    charstring          SessionId   optional

  };


	  template (value) IMS_PTC_DialogInfo_Type cs_IMS_DialogInfo_Init :=

  {  /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT) */

    MobileOriginated := omit,

    RemoteTag := omit,

    RemoteURI := omit,

    RemoteCseq := omit,           /* @sic R5s140235: changed to omit sic@ */

    LocalTag := omit,

    LocalURI := omit,

    LocalCseq := 0,

    CallId := omit,

    SessionId := omit

  };


	  function f_IMS_INVITE_ReceiveRequest_Common(INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                              template (present) MessageHeader p_MessageHeader := cr_MessageHeader_Dummy,

                                              OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                              template (present) MessageBody p_MessageBody,

                                              template (omit) charstring p_OtherCalleeUri := omit,

                                              template (omit) float p_WaitDuration := omit,

                                              boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                              boolean p_A8 := pc_IMS_Geolocation,

                                              boolean p_A9 := false,

                                              boolean p_A10 := false,

                                              boolean p_A11 := false,

                                              boolean p_A12 := false)

    runs on IMS_PTC return template (omit) IMS_DATA_REQ

  {
...

    alt {

      [] IMS_Server.receive(car_IMS_Invite_Request(cr_INVITE_Request(v_ExpectedCalleeUri, v_MessageHeader_Invite, p_MessageBody))) -> value v_IMS_DATA_REQ

        {

          t_Timer.stop;

          if (p_Context == A_2_1_A6) {      // emergency call without registration   @sic R5s130756 change 14 sic@

            f_IMS_PTC_ImsInfo_Init(v_IMS_DATA_REQ.RoutingInfo.UE_Address, v_IMS_DATA_REQ.RoutingInfo.NW_Address);

          }

          f_IMS_A_2_1_Invite_CommonChecks(v_IMS_DATA_REQ, p_Context, p_A3, p_A8, p_A10, p_A11, p_A12);

          if (p_Context != A_2_1_A5) {      /* @sic R5s140020 change 8.2 sic@ */

            f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_IMS_DATA_REQ.Request.Invite.msgHeader);

          } else {
            if (not isvalue(f_IMS_PTC_ImsInfo_CseqGet(dialogRemote))) {  /* @sic R5s140235 sic@ */

              f_IMS_PTC_ImsInfo_CseqSet(dialogRemote, v_IMS_DATA_REQ.Request.Invite.msgHeader.cSeq.seqNumber);
            }
          }
          return v_IMS_DATA_REQ;

        }

      [v_TimerControl] t_Timer.timeout {}

    }
...

  }

